Code should execute sequentially if run in a Jupyter notebook

# Job Search I: The McCall Search Model¶

“Questioning a McCall worker is like having a conversation with an out-of-work friend: ‘Maybe you are setting your sights too high’, or ‘Why did you quit your old job before you had a new one lined up?’ This is real social science: an attempt to model, to understand, human behavior by visualizing the situation people find themselves in, the options they face and the pros and cons as they themselves see them.” – Robert E. Lucas, Jr.

## Overview¶

The McCall search model [McC70] helped transform economists’ way of thinking about labor markets

To clarify vague notions such as “involuntary” unemployment, McCall modeled the decision problem of unemployed agents directly, in terms of factors such as

• current and likely future wages
• impatience
• unemployment compensation

To solve the decision problem he used dynamic programming

Here we set up McCall’s model and adopt the same solution method

As we’ll see, McCall’s model is not only interesting in its own right but also an excellent vehicle for learning dynamic programming

## The McCall Model¶

An unemployed worker receives in each period a job offer at wage $$W_t$$

At time $$t$$, our worker has two choices:

1. Accept the offer and work permanently at constant wage $$W_t$$
2. Reject the offer, receive unemployment compensation $$c$$, and reconsider next period

The wage sequence $$\{W_t\}$$ is assumed to be iid with probability mass function $$p_1, \ldots, p_n$$

Here $$p_i$$ is the probability of observing wage offer $$W_t = w_i$$ in the set $$w_1, \ldots, w_n$$

The worker is infinitely lived and aims to maximize the expected discounted sum of earnings

$\mathbb{E} \sum_{t=0}^{\infty} \beta^t Y_t$

The constant $$\beta$$ lies in $$(0, 1)$$ and is called a discount factor

The smaller is $$\beta$$, the more the worker discounts future utility relative to current utility

The variable $$Y_t$$ is income, equal to

• his wage $$W_t$$ when employed
• unemployment compensation $$c$$ when unemployed

• Waiting too long for a good offer is costly, since the future is discounted
• Accepting too early is costly, since better offers might arrive in the future

To decide optimally in the face of this trade off, we use dynamic programming

Dynamic programming can be thought of as a two step procedure that

1. first assigns values to “states” and
2. then deduces optimal actions given those values

We’ll go through these steps in turn

### The Value Function¶

In order to optimally trade off current and future rewards, we need to think about two things:

1. the current payoffs we get from different choices
2. the different states that those choices will lead to next period (in this case, either employment or unemployment)

To weigh these two aspects of the decision problem, we need to assign values to states

To this end, let $$V(w)$$ be the total lifetime value accruing to an unemployed worker who enters the current period unemployed but with wage offer $$w$$ in hand

More precisely, $$V(w)$$ denotes the value of the objective function (1) when an agent in this situation makes optimal decisions now and at all future points in time

Of course $$V(w)$$ is not trivial to calculate because we don’t yet know what decisions are optimal and what aren’t!

But think of $$V$$ as a function that assigns to each possible wage $$w$$ the maximal lifetime value that can be obtained with that offer in hand

A crucial observation is that this function $$V$$ must satisfy the recursion

(1)$V(w) = \max \left\{ \frac{w}{1 - \beta}, \, c + \beta \sum_{i=1}^n V(w_i) p_i \right\}$

for every possible $$w_i$$ in $$w_1, \ldots, w_n$$

This important equation is a version of the Bellman equation, which is ubiquitous in economic dynamics and other fields involving planning over time

The intuition behind it is as follows:

• the first term inside the max operation is the lifetime payoff from accepting current offer $$w$$, since
$w + \beta w + \beta^2 w + \cdots = \frac{w}{1 - \beta}$
• the second term inside the max operation is the continuation value, which is the lifetime payoff from rejecting the current offer and then behaving optimally in all subsequent periods

If we optimize and pick the best of these two options, we obtain maximal lifetime value from today, given current offer $$w$$

But this is precisely $$V(w)$$, which is the l.h.s. of (1)

### The Optimal Policy¶

Suppose for now that we are able to solve (1) for the unknown function $$V$$

Once we have this function in hand we can behave optimally (i.e., make the right choice between accept and reject)

All we have to do is select the maximal choice on the r.h.s. of (1)

The optimal action is best thought of as a policy, which is, in general, a map from states to actions

In our case, the state is the current wage offer $$w$$

Given any $$w$$, we can read off the corresponding best choice (accept or reject) by picking the max on the r.h.s. of (1)

Thus, we have a map from $$\RR$$ to $$\{0, 1\}$$, with 1 meaning accept and zero meaning reject

We can write the policy as follows

$\sigma(w) := \mathbf{1} \left\{ \frac{w}{1 - \beta} \geq c + \beta \sum_{i=1}^n V(w_i) p_i \right\}$

Here $$\mathbf{1}\{ P \} = 1$$ if statement $$P$$ is true and equals zero otherwise

We can also write this as

$\sigma(w) := \mathbf{1} \{ w \geq \bar w \}$

where

$\bar w := (1 - \beta) \left\{ c + \beta \sum_{i=1}^n V(w_i) p_i \right\}$

Here $$\bar w$$ is a constant depending on $$\beta, c$$ and the wage distribution, called the reservation wage

The agent should accept if and only if the current wage offer exceeds the reservation wage

Clearly, we can compute this reservation wage if we can compute the value function

## Computing the Optimal Policy: Take 1¶

To put the above ideas into action, we need to compute the value function at points $$w_1, \ldots, w_n$$

In doing so, we can identify these values with the vector $$v = (v_i)$$ where $$v_i := V(w_i)$$

In view of (1), this vector satisfies the nonlinear system of equations

(2)$v_i = \max \left\{ \frac{w_i}{1 - \beta}, \, c + \beta \sum_{i=1}^n v_i p_i \right\} \quad \text{for } i = 1, \ldots, n$

It turns out that there is exactly one vector $$v := (v_i)_{i=1}^n$$ in $$\mathbb R^n$$ that satisfies this equation

### The Algorithm¶

To compute this vector, we proceed as follows:

Step 1: pick an arbitrary initial guess $$v \in \mathbb R^n$$

Step 2: compute a new vector $$v' \in \mathbb R^n$$ via

(3)$v'_i = \max \left\{ \frac{w_i}{1 - \beta}, \, c + \beta \sum_{i=1}^n v_i p_i \right\} \quad \text{for } i = 1, \ldots, n$

Step 3: calculate a measure of the deviation between $$v$$ and $$v'$$, such as $$\max_i |v_i - v_i'|$$

Step 4: if the deviation is larger than some fixed tolerance, set $$v = v'$$ and go to step 2, else continue

Step 5: return $$v$$

This algorithm returns an arbitrarily good approximation to the true solution to (2), which represents the value function

(Arbitrarily good means here that the approximation converges to the true solution as the tolerance goes to zero)

### The Fixed Point Theory¶

What’s the math behind these ideas?

First, one defines a mapping $$T$$ from $$\mathbb R^n$$ to itself via

(4)$Tv_i = \max \left\{ \frac{w_i}{1 - \beta}, \, c + \beta \sum_{i=1}^n v_i p_i \right\} \quad \text{for } i = 1, \ldots, n$

(A new vector $$Tv$$ is obtained from given vector $$v$$ by evaluating the r.h.s. at each $$i$$)

One can show that the conditions of the Banach contraction mapping theorem are satisfied by $$T$$ as a self-mapping on $$\RR^n$$

One implication is that $$T$$ has a unique fixed point in $$\mathbb R^n$$

Moreover, it’s immediate from the definition of $$T$$ that this fixed point is precisely the value function

The iterative algorithm presented above corresponds to iterating with $$T$$ from some initial guess $$v$$

The Banach contraction mapping theorem tells us that this iterative process generates a sequence that converges to the fixed point

### Implementation¶

using PyPlot
using QuantEcon
using Distributions
using LaTeXStrings
plt = PyPlot


Here’s the distribution of wage offers we’ll work with

n, a, b = 50, 200, 100
w_min, w_max = 10, 60
const w_vals = linspace(w_min, w_max, n+1)
dist = BetaBinomial(n, a, b)
const p_vals = pdf.(dist, support(dist))

fig, ax = plt.subplots(figsize=(9, 6.5))
ax[:stem](w_vals, p_vals, label=L"$p_i$")
ax[:set_xlabel]("wages")
ax[:set_ylabel]("probabilities")

plt.show()


First let’s have a look at the sequence of approximate value functions that the algorithm above generates

Default parameter values are embedded in the function

Our initial guess $$v$$ is the value of accepting at every given wage

function plot_value_function_seq(ax,
c=25,
β=0.99,
num_plots=6)

v = w_vals ./ (1 - β)
v_next = similar(v)

for i in 1:num_plots
ax[:plot](w_vals, v, label="iterate $i") # Update guess for (j, w) in enumerate(w_vals) stop_val = w / (1 - β) cont_val = c + β * sum(v .* p_vals) v_next[j] = max(stop_val, cont_val) end v[:] = v_next end ax[:legend](loc="lower right") end fig, ax = plt.subplots(figsize=(9, 6.5)) plot_value_function_seq(ax) plt.show()  Here’s more serious iteration effort, that continues until measured deviation between successive iterates is below tol function compute_reservation_wage(c::Float64, β::Float64; max_iter::Int64=500, tol::Float64=1e-6) # == First compute the value function == # v = w_vals ./ (1 - β) v_next = similar(v) i = 0 error = tol + 1 while i < max_iter && error > tol for (j, w) in enumerate(w_vals) stop_val = w / (1 - β) cont_val = c + β * sum(v .* p_vals) v_next[j] = max(stop_val, cont_val) end error = maximum(abs.(v_next - v)) i += 1 v[:] = v_next # copy contents into v end # == Now compute the reservation wage == # return (1 - β) * (c + β * sum(v .* p_vals)) end  Let’s compute the reservation wage at the default parameters c = 25.0 β = 0.99 compute_reservation_wage(c, β)  47.316499709964695  ### Comparative Statics¶ Now we know how to compute the reservation wage, let’s see how it varies with parameters In particular, let’s look at what happens when we change $$\beta$$ and $$c$$ grid_size = 25 R = rand((grid_size, grid_size)) c_vals = linspace(10.0, 30.0, grid_size) β_vals = linspace(0.9, 0.99, grid_size) for (i, c) in enumerate(c_vals) for (j, β) in enumerate(β_vals) R[i, j] = compute_reservation_wage(c, β) end end  fig, ax = plt.subplots(figsize=(10, 5.7)) cs1 = ax[:contourf](c_vals, β_vals, R', alpha=0.75) ctr1 = ax[:contour](c_vals, β_vals, R') plt.clabel(ctr1, inline=1, fontsize=13) plt.colorbar(cs1, ax=ax) ax[:set_title]("reservation wage") ax[:set_xlabel](L"$c$", fontsize=16) ax[:set_ylabel](L"$β\$", fontsize=16)

ax[:ticklabel_format](useOffset=false)

plt.show()


As expected, the reservation wage increases both with patience and with unemployment compensation

## Computing the Optimal Policy: Take 2¶

The approach to dynamic programming just described is very standard and broadly applicable

For this particular problem, there’s also an easier way, which circumvents the need to compute the value function

Let $$\psi$$ denote the value of not accepting a job in this period but then behaving optimally in all subsequent periods

That is,

(5)$\psi = c + \beta \sum_{i=1}^n V(w_i) p_i \quad$

where $$V$$ is the value function

By the Bellman equation, we then have

$V(w_i) = \max \left\{ \frac{w_i}{1 - \beta}, \, \psi \right\}$

Substituting this last equation into (5) gives

(6)$\psi = c + \beta \sum_{i=1}^n \max \left\{ \frac{w_i}{1 - \beta}, \psi \right\} p_i \quad$

This is a nonlinear equation that we can solve for $$\psi$$

The natural solution method for this kind of nonlinear equation is iterative

That is,

Step 1: pick an initial guess $$\psi$$

Step 2: compute the update $$\psi'$$ via

(7)$\psi' = c + \beta \sum_{i=1}^n \max \left\{ \frac{w_i}{1 - \beta}, \psi \right\} p_i \quad$

Step 3: calculate the deviation $$|\psi - \psi'|$$

Step 4: if the deviation is larger than some fixed tolerance, set $$\psi = \psi'$$ and go to step 2, else continue

Step 5: return $$\psi$$

Once again, one can use the Banach contraction mapping theorem to show that this process always converges

The big difference here, however, is that we’re iterating on a single number, rather than an $$n$$-vector

Here’s an implementation:

function compute_reservation_wage_two(c::Float64;
β=0.99,
max_iter=500,
tol=1e-5)
# == First compute ψ == #

ψ = dot(w_vals, p_vals) ./ (1 - β)
i = 0
error = tol + 1
while i < max_iter && error > tol

s = max.((w_vals ./ (1 - β)), ψ)
ψ_next = c + β * dot(s, p_vals)

error = abs.(ψ_next - ψ)
i += 1

ψ = ψ_next
end

# == Now compute the reservation wage == #

return (1 - β) * (c + β * ψ)
end


You can use this code to solve the exercise below

## Exercises¶

### Exercise 1¶

Compute the average duration of unemployment when $$\beta=0.99$$ and $$c$$ takes the following values

c_vals = np.linspace(10, 40, 25)

That is, start the agent off as unemployed, computed their reservation wage given the parameters, and then simulate to see how long it takes to accept

Repeat a large number of times and take the average

Plot mean unemployment duration as a function of $$c$$ in c_vals

## Solutions¶

### Exercise 1¶

Here’s one solution

function compute_stopping_time(w_bar; seed=1234)

srand(seed)
stopping_time = 0

t = 1
while true
# Generate a wage draw
d = DiscreteRV(p_vals)
w = w_vals[rand(d)]
if w >= w_bar
stopping_time = t
break
else
t += 1
end
end
return stopping_time
end

function compute_mean_stopping_time(w_bar, num_reps=10000)
obs = zeros(num_reps)
for i in 1:num_reps
obs[i] = compute_stopping_time(w_bar, seed=i)
end
return mean(obs)
end

c_vals = linspace(10, 40, 25)
stop_times = similar(c_vals)

for (i, c) in enumerate(c_vals)
w_bar = compute_reservation_wage_two(c)
stop_times[i] = compute_mean_stopping_time(w_bar)
end

fig, ax = plt.subplots(figsize=(9, 6.5))

ax[:plot](c_vals, stop_times, label="mean unemployment duration")
ax[:set](xlabel="unemployment compensation", ylabel="months")
ax[:legend]()

plt.show()

• Share page